

# NAG Library Routine Document

## F08GNF (ZHPEV)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F08GNF (ZHPEV) computes all the eigenvalues and, optionally, all the eigenvectors of a complex  $n$  by  $n$  Hermitian matrix  $A$  in packed storage.

### 2 Specification

```
SUBROUTINE F08GNF (JOBZ, UPLO, N, AP, W, Z, LDZ, WORK, RWORK, INFO)
INTEGER                N, LDZ, INFO
REAL (KIND=nag_wp)    W(N), RWORK(3*N-2)
COMPLEX (KIND=nag_wp) AP(*), Z(LDZ,*), WORK(2*N-1)
CHARACTER(1)          JOBZ, UPLO
```

The routine may be called by its LAPACK name *zhpev*.

### 3 Description

The Hermitian matrix  $A$  is first reduced to real tridiagonal form, using unitary similarity transformations, and then the  $QR$  algorithm is applied to the tridiagonal matrix to compute the eigenvalues and (optionally) the eigenvectors.

### 4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia <http://www.netlib.org/lapack/lug>

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

### 5 Parameters

- 1: JOBZ – CHARACTER(1) *Input*  
*On entry:* indicates whether eigenvectors are computed.  
 JOBZ = 'N'  
     Only eigenvalues are computed.  
 JOBZ = 'V'  
     Eigenvalues and eigenvectors are computed.  
*Constraint:* JOBZ = 'N' or 'V'.
- 2: UPLO – CHARACTER(1) *Input*  
*On entry:* if UPLO = 'U', the upper triangular part of  $A$  is stored.  
 If UPLO = 'L', the lower triangular part of  $A$  is stored.  
*Constraint:* UPLO = 'U' or 'L'.

- 3: N – INTEGER *Input*  
*On entry:*  $n$ , the order of the matrix  $A$ .  
*Constraint:*  $N \geq 0$ .
- 4: AP(\*) – COMPLEX (KIND=nag\_wp) array *Input/Output*  
**Note:** the dimension of the array AP must be at least  $\max(1, N \times (N + 1)/2)$ .  
*On entry:* the upper or lower triangle of the  $n$  by  $n$  Hermitian matrix  $A$ , packed by columns.  
 More precisely,  
   if UPLO = 'U', the upper triangle of  $A$  must be stored with element  $A_{ij}$  in  
   AP( $i + j(j - 1)/2$ ) for  $i \leq j$ ;  
   if UPLO = 'L', the lower triangle of  $A$  must be stored with element  $A_{ij}$  in  
   AP( $i + (2n - j)(j - 1)/2$ ) for  $i \geq j$ .  
*On exit:* AP is overwritten by the values generated during the reduction to tridiagonal form. The elements of the diagonal and the off-diagonal of the tridiagonal matrix overwrite the corresponding elements of  $A$ .
- 5: W(N) – REAL (KIND=nag\_wp) array *Output*  
*On exit:* the eigenvalues in ascending order.
- 6: Z(LDZ,\*) – COMPLEX (KIND=nag\_wp) array *Output*  
**Note:** the second dimension of the array Z must be at least  $\max(1, N)$  if JOBZ = 'V', and at least 1 otherwise.  
*On exit:* if JOBZ = 'V', Z contains the orthonormal eigenvectors of the matrix  $A$ , with the  $i$ th column of Z holding the eigenvector associated with  $W(i)$ .  
 If JOBZ = 'N', Z is not referenced.
- 7: LDZ – INTEGER *Input*  
*On entry:* the first dimension of the array Z as declared in the (sub)program from which F08GNF (ZHPEV) is called.  
*Constraints:*  
   if JOBZ = 'V',  $LDZ \geq \max(1, N)$ ;  
   otherwise  $LDZ \geq 1$ .
- 8: WORK( $2 \times N - 1$ ) – COMPLEX (KIND=nag\_wp) array *Workspace*
- 9: RWORK( $3 \times N - 2$ ) – REAL (KIND=nag\_wp) array *Workspace*
- 10: INFO – INTEGER *Output*  
*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

INFO < 0

If INFO =  $-i$ , argument  $i$  had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO =  $i$ , the algorithm failed to converge;  $i$  off-diagonal elements of an intermediate tridiagonal form did not converge to zero.

## 7 Accuracy

The computed eigenvalues and eigenvectors are exact for a nearby matrix  $(A + E)$ , where

$$\|E\|_2 = O(\epsilon)\|A\|_2,$$

and  $\epsilon$  is the *machine precision*. See Section 4.7 of Anderson *et al.* (1999) for further details.

## 8 Parallelism and Performance

F08GNF (ZHPEV) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F08GNF (ZHPEV) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

Each eigenvector is normalized so that the element of largest absolute value is real and positive.

The total number of floating-point operations is proportional to  $n^3$ .

The real analogue of this routine is F08GAF (DSPEV).

## 10 Example

This example finds all the eigenvalues of the Hermitian matrix

$$A = \begin{pmatrix} 1 & 2 - i & 3 - i & 4 - i \\ 2 + i & 2 & 3 - 2i & 4 - 2i \\ 3 + i & 3 + 2i & 3 & 4 - 3i \\ 4 + i & 4 + 2i & 4 + 3i & 4 \end{pmatrix},$$

together with approximate error bounds for the computed eigenvalues.

### 10.1 Program Text

```

Program f08gnfe

!      F08GNF Example Program Text

!      Mark 25 Release. NAG Copyright 2014.

!      .. Use Statements ..
      Use nag_library, Only: nag_wp, x02ajf, zhpev
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
      Character (1), Parameter    :: uplo = 'U'
!      .. Local Scalars ..
      Real (Kind=nag_wp)          :: eerrbd, eps
      Integer                     :: i, info, j, n
!      .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: ap(:), work(:)

```

```

Complex (Kind=nag_wp)          :: dummy(1,1)
Real (Kind=nag_wp), Allocatable :: rwork(:), w(:)
! .. Intrinsic Procedures ..
Intrinsic                      :: abs, max
! .. Executable Statements ..
Write (nout,*) 'F08GNF Example Program Results'
Write (nout,*)
! Skip heading in data file
Read (nin,*)
Read (nin,*) n

Allocate (ap((n*(n+1))/2),work(2*n-1),rwork(3*n-2),w(n))

! Read the upper or lower triangular part of the matrix A from
! data file

If (uplo=='U') Then
  Read (nin,*)((ap(i+(j*(j-1))/2),j=i,n),i=1,n)
Else If (uplo=='L') Then
  Read (nin,*)((ap(i+((2*n-j)*(j-1))/2),j=1,i),i=1,n)
End If

! Solve the Hermitian eigenvalue problem

! The NAG name equivalent of zhpev is f08gnf
Call zhpev('No vectors',uplo,n,ap,w,dummy,1,work,rwork,info)

If (info==0) Then

!   Print solution

  Write (nout,*) 'Eigenvalues'
  Write (nout,99999) w(1:n)

!   Get the machine precision, EPS and compute the approximate
!   error bound for the computed eigenvalues. Note that for
!   the 2-norm, max( abs(W(i)) ) = norm(A), and since the
!   eigenvalues are returned in ascending order
!   max( abs(W(i)) ) = max( abs(W(1)), abs(W(n))

  eps = x02ajf()
  eerrbd = eps*max(abs(w(1)),abs(w(n)))

!   Print the approximate error bound for the eigenvalues

  Write (nout,*)
  Write (nout,*) 'Error estimate for the eigenvalues'
  Write (nout,99998) eerrbd
Else
  Write (nout,99997) 'Failure in ZHPEV. INFO =', info
End If

99999 Format (3X,(8F8.4))
99998 Format (4X,1P,6E11.1)
99997 Format (1X,A,I4)
End Program f08gnfe

```

## 10.2 Program Data

F08GNF Example Program Data

```

4                                     :Value of N
(1.0, 0.0) (2.0, -1.0) (3.0, -1.0) (4.0, -1.0)
          (2.0, 0.0) (3.0, -2.0) (4.0, -2.0)
                    (3.0, 0.0) (4.0, -3.0)
                              (4.0, 0.0) :End of matrix A

```

### 10.3 Program Results

F08GNF Example Program Results

Eigenvalues

-4.2443 -0.6886 1.1412 13.7916

Error estimate for the eigenvalues

1.5E-15

---